Code should execute sequentially if run in a Jupyter notebook

# History Dependent Public Policies¶

## Overview¶

This lecture describes history-dependent public policies and some of their representations

History dependent policies are decision rules that depend on the entire past history of the state variables

History dependent policies naturally emerge in Ramsey problems

A Ramsey planner (typically interpreted as a government) devises a plan of actions at time $$t=0$$ to follow at all future dates and for all contingencies

In order to make a plan, he takes as given Euler equations expressing private agents’ first-order necessary conditions

He also takes into account that his future actions affect earlier decisions by private agents, an avenue opened up by the maintained assumption of rational expectations

Another setting in which history dependent policies naturally emerge is where instead of a Ramsey planner there is a sequence of government administrators whose time $$t$$ member takes as given the policies used by its successors

We study these ideas in the context of a model in which a benevolent tax authority is forced

• to raise a prescribed present value of revenues
• to do so by imposing a distorting flat rate tax on the output of a competitive representative firm

The firm faces costs of adjustment and lives within a competitive equilibrium, which in turn imposes restrictions on the tax authority [1]

### References¶

The presentation below is based on a recent paper by Evans and Sargent [ES13]

Regarding techniques, we will make use of the methods described in

## Two Sources of History Dependence¶

We compare two timing protocols

1. An infinitely lived benevolent tax authority solves a Ramsey problem
2. There is a sequence of tax authorities, each choosing only a time $$t$$ tax rate

Under both timing protocols, optimal tax policies are history-dependent

But history dependence captures different economic forces across the two timing protocols

In the first timing protocol, history dependence expresses the time-inconsistency of the Ramsey plan

In the second timing protocol, history dependence reflects the unfolding of constraints that assure that a time $$t$$ government administrator wants to confirm the representative firm’s expectations about government actions

We describe recursive representations of history-dependent tax policies under both timing protocols

### Ramsey Timing Protocol¶

The first timing protocol models a policy maker who can be said to ‘commit’, choosing a sequence of tax rates once-and-for-all at time $$0$$

### Sequence of Governments Timing Protocol¶

For the second timing protocol we use the notion of a sustainable plan proposed in [CK90], also referred to as a credible public policy in [Sto89]

A key idea here is that history-dependent policies can be arranged so that, when regarded as a representative firm’s forecasting functions, they confront policy makers with incentives to confirm them

We follow Chang [Cha98] in expressing such history-dependent plans recursively

Credibility considerations contribute an additional auxiliary state variable in the form of a promised value to the planner

It expresses how decisions must unfold to give the government the incentive to confirm private sector expectations when the government chooses sequentially

Note

We occasionally hear confusion about the consequences of recursive representations of government policies under our two timing protocols. It is incorrect to regard a recursive representation of the Ramsey plan as in any way ‘solving a time-inconsistency problem’. On the contrary, the evolution of the auxiliary state variable that augments the authentic ones under our first timing protocol ought to be viewed as expressing the time-inconsistency of a Ramsey plan. Despite that, in literatures about practical monetary policy one sometimes hears interpretations that sell Ramsey plans in settings where our sequential timing protocol is the one that more accurately characterizes decision making. Please beware of discussions that toss around claims about credibility if you don’t also see recursive representations of policies with the complete list of state variables appearing in our [Cha98] -like analysis that we present below.

## Competitive equilibrium¶

A representative competitive firm sells output $$q_t$$ at price $$p_t$$ when market-wide output is $$Q_t$$

The market as a whole faces a downward sloping inverse demand function

(1)$p_t = A_0 - A_1 Q_t, \quad A_0 >0, A_1 >0$

The representative firm

• has given initial condition $$q_0$$
• endures quadratic adjustment costs $$\frac{d}{2} (q_{t+1} - q_t)^2$$
• pays a flat rate tax $$\tau_t$$ per unit of output
• treats $$\{p_t, \tau_t\}_{t=0}^\infty$$ as exogenous
• chooses $$\{q_{t+1}\}_{t=0}^\infty$$ to maximize
(2)$\sum_{t=0}^\infty \beta^t \bigl\{ p_t q_t - \frac{d}{2}(q_{t+1} - q_t)^2 - \tau_t q_t \bigr\}$

Let $$u_t := q_{t+1} - q_t$$ be the firm’s ‘control variable’ at time $$t$$

First-order conditions for the representative firm’s problem are

(3)$u_t = \frac{\beta}{d} p_{t+1} + \beta u_{t+1} - \frac{\beta}{d} \tau_{t+1}, \quad t = 0, 1, \ldots$

To compute a competitive equilibrium, it is appropriate to take (3), eliminate $$p_t$$ in favor of $$Q_t$$ by using (1), and then set $$q_t = Q_t$$

This last step makes the representative firm be representative [2]

We arrive at

(4)$u_t = \frac{\beta}{d} ( A_0 - A_1 Q_{t+1} ) + \beta u_{t+1} - \frac{\beta}{d} \tau_{t+1}$
(5)$Q_{t+1} = Q_t + u_t$

Notation: For any scalar $$x_t$$, let $$\vec x = \{x_t\}_{t=0}^\infty$$

Given a tax sequence $$\{\tau_{t+1}\}_{t=0}^\infty$$, a competitive equilibrium is a price sequence $$\vec p$$ and an output sequence $$\vec Q$$ that satisfy (1), (4), and (5)

For any sequence $$\vec x = \{x_t\}_{t=0}^\infty$$, the sequence $$\vec x_1 := \{x_t\}_{t=1}^\infty$$ is called the continuation sequence or simply the continuation

Note that a competitive equilibrium consists of a first period value $$u_0 = Q_1-Q_0$$ and a continuation competitive equilibrium with initial condition $$Q_1$$

Also, a continuation of a competitive equilibrium is a competitive equilibrium

Following the lead of [Cha98], we shall make extensive use of the following property:

• A continuation $$\vec \tau_1 = \{\tau_{t}\}_{t=1}^\infty$$ of a tax policy $$\vec \tau$$ influences $$u_0$$ via (4) entirely through its impact on $$u_1$$

A continuation competitive equilibrium can be indexed by a $$u_1$$ that satisfies (4)

In the spirit of [KP80a] , we shall use $$u_{t+1}$$ to describe what we shall call a promised marginal value that a competitive equilibrium offers to a representative firm [3]

Define $$Q^t := [Q_0, \ldots, Q_t]$$

A history-dependent tax policy is a sequence of functions $$\{\sigma_t\}_{t=0}^\infty$$ with $$\sigma_t$$ mapping $$Q^t$$ into a choice of $$\tau_{t+1}$$

Below, we shall

• Study history-dependent tax policies that either solve a Ramsey plan or are credible
• Describe recursive representations of both types of history-dependent policies

## Ramsey Problem¶

The planner’s objective is cast in terms of consumer surplus net of the firm’s adjustment costs

Consumer surplus is

$\int_0^Q ( A_0 - A_1 x) dx = A_0 Q - \frac{A_1}{2} Q^2$

Hence the planner’s one-period return function is

(6)$A_0 Q_t - \frac{A_1}{2} Q_t^2 - \frac{d}{2} u_t^2$

At time $$t=0$$, a Ramsey planner faces the intertemporal budget constraint

(7)$\sum_{t=1}^\infty \beta^t \tau_t Q_t = G_0$

Note that (7) forbids taxation of initial output $$Q_0$$

The Ramsey problem is to choose a tax sequence $$\vec \tau_1$$ and a competitive equilibrium outcome $$(\vec Q, \vec u)$$ that maximize

(8)$\sum_{t=0}^\infty \beta^t \left\{ A_0 Q_t - \frac{A_1}{2}Q_t^2 - \frac{d}{2} u_t^2 \right\}$

subject to (7)

Thus, the Ramsey timing protocol is:

1. At time $$0$$, knowing $$(Q_0, G_0)$$, the Ramsey planner chooses $$\{\tau_{t+1}\}_{t=0}^\infty$$
2. Given $$\bigl(Q_0, \{\tau_{t+1}\}_{t=0}^\infty\bigr)$$, a competitive equilibrium outcome $$\{u_t, Q_{t+1}\}_{t=0}^\infty$$ emerges

Note

In bringing out the timing protocol associated with a Ramsey plan, we run head on into a set of issues analyzed by Bassetto [Bas05]. This is because our definition of the Ramsey timing protocol doesn’t completely describe all conceivable actions by the government and firms as time unfolds. For example, the definition is silent about how the government would respond if firms, for some unspecified reason, were to choose to deviate from the competitive equilibrium associated with the Ramsey plan, possibly prompting violation of government budget balance. This is an example of the issues raised by [Bas05], who identifies a class of government policy problems whose proper formulation requires supplying a complete and coherent description of all actors’ behavior across all possible histories. Implicitly, we are assuming that a more complete description of a government strategy could be specified that (a) agrees with ours along the Ramsey outcome, and (b) suffices uniquely to implement the Ramsey plan by deterring firms from taking actions that deviate from the Ramsey outcome path.

### Computing a Ramsey Plan¶

The planner chooses $$\{u_t\}_{t=0}^\infty, \{\tau_t\}_{t=1}^\infty$$ to maximize (8) subject to (4), (5), and (7)

To formulate this problem as a Lagrangian, attach a Lagrange multiplier $$\mu$$ to the budget constraint (7)

Then the planner chooses $$\{u_t\}_{t=0}^\infty, \{\tau_t\}_{t=1}^\infty$$ to maximize and the Lagrange multiplier $$\mu$$ to minimize

(9)$\sum_{t=0}^\infty \beta^t ( A_0 Q_t - \frac{A_1}{2}Q_t^2 - \frac{d}{2} u_t^2 ) + \mu \left[ \sum_{t=0}^\infty \beta^t \tau_t Q_t -G_0 - \tau_0 Q_0 \right]$

subject to and (4) and (5)

The Ramsey problem is a special case of the linear quadratic dynamic Stackelberg problem analyzed in this lecture

The key implementability conditions are (4) for $$t \geq 0$$

Holding fixed $$\mu$$ and $$G_0$$, the Lagrangian for the planning problem can be abbreviated as

$\max_{\{u_t, \tau_{t+1}\}} \sum_{t=0}^\infty \beta^t \left\{ A_0 Q_t-\frac {A_1}2 Q_t^2-\frac d2 u_t^2+\mu \tau_t Q_t \right\}$

Define

$\begin{split} z_t := \begin{bmatrix} 1 \\ Q_t \\ \tau_t \end{bmatrix} \quad \text{and} \quad y_t := \begin{bmatrix} z_t \\ u_t \end{bmatrix} = \begin{bmatrix} 1 \\ Q_t \\ \tau_t \\ u_t \end{bmatrix}\end{split}$

Here the elements of $$z_t$$ are natural state variables and $$u_t$$ is a forward looking variable that we treat as a state variable for $$t \geq 1$$

But $$u_0$$ is a choice variable for the Ramsey planner.

We include $$\tau_t$$ as a state variable for bookkeeping purposes: it helps to map the problem into a linear regulator problem with no cross products between states and controls

However, it will be a redundant state variable in the sense that the optimal tax $$\tau_{t+1}$$ will not depend on $$\tau_t$$

The government chooses $$\tau_{t+1}$$ at time $$t$$ as a function of the time $$t$$ state

Thus, we can rewrite the Ramsey problem as

(10)$\max_{\{y_t, \tau_{t+1}\}} -\sum_{t=0}^\infty \beta^t y_t' Ry_t$

subject to $$z_0$$ given and the law of motion

(11)$y_{t+1} = A y_t + B \tau_{t+1}$

where

$\begin{split}R = \begin{bmatrix} 0 &-\frac{A_0}{2} & 0 & 0 \\ -\frac{A_0}{2} & \frac{A_1}{2} & \frac {-\mu}{2} & 0 \\ 0 & \frac{-\mu}{2} & 0 & 0 \\ 0 & 0 & 0 & \frac{d}{2} \end{bmatrix}, \quad A = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 1 \\ 0 & 0 & 0 & 0 \\ -\frac{A_0}{d} & \frac{A_1}{d} & 0 & \frac{A_1}{d} + \frac{1}{\beta} \end{bmatrix}, \quad B = \begin{bmatrix} 0 \\ 0 \\ 1 \\ \frac{1}{d} \end{bmatrix}\end{split}$

## Two Subproblems¶

Working backwards, we first present the Bellman equation for the value function that takes both $$z_t$$ and $$u_t$$ as given. Then we present a value function that takes only $$z_0$$ as given and is the indirect utility function that arises from choosing $$u_0$$ optimally.

Let $$v(Q_t, \tau_t, u_t)$$ be the optimum value function for the time $$t \geq 1$$ government administrator facing state $$Q_t, \tau_t, u_t$$.

Let $$w(Q_0)$$ be the value of the Ramsey plan starting from $$Q_0$$

### Subproblem 1¶

Here the Bellman equation is

$v(Q_t,\tau_t,u_t) = \max_{\tau_{t+1}} \left\{ A_0 Q_t-\frac {A_1}2 Q_t^2-\frac d2 u_t^2+\mu\tau_tQ_t + \beta v(Q_{t+1},\tau_{t+1},u_{t+1}) \right\}$

where the maximization is subject to the constraints

$Q_{t+1} = Q_t+u_t$

and

$u_{t+1} = -\frac{A_0}d+\frac{A_1}d Q_t + \frac{A_1}d+\frac1\beta u_t+\frac1d \tau_{t+1}$

Here we regard $$u_t$$ as a state

### Subproblem 2¶

The subproblem 2 Bellman equation is

$w(z_0) = \max_{u_0} v (Q_0,0, u_0)$

### Details¶

Define the state vector to be

$\begin{split}y_t = \begin{bmatrix} 1 \\ Q_t \\ \tau_t \\ u_t \end{bmatrix} = \begin{bmatrix} z_t \\ u_t \end{bmatrix},\end{split}$

where $$z_t = \begin{bmatrix} 1 & Q_t & \tau_t\end{bmatrix}'$$ are authentic state variables and $$u_t$$ is a variable whose time $$0$$ value is a ‘jump’ variable but whose values for dates $$t \geq 1$$ will become state variables that encode history dependence in the Ramsey plan

(12)$v(y_t) = \max_{\tau_{t+1}} \left\{ -y_t'Ry_t+\beta v(y_{t+1}) \right\}$

where the maximization is subject to the constraint

$y_{t+1} = Ay_t+B\tau_{t+1}$

and where

$\begin{split}R = \begin{bmatrix} 0 & -\frac {A_0}2 & 0 & 0 \\ -\frac{A_0}2 & \frac{A_1}2 & \frac {-\mu}{2}&0\\ 0 & \frac{-\mu}{2}&0 & 0 \\ 0 & 0 & 0&\frac d2\end{bmatrix}, \: A = \begin{bmatrix}1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 1\\ 0 & 0 & 0 & 0 \\ -\frac{A_0}d & \frac{A_1}d & 0 & \frac{A_1}d+\frac1\beta\end{bmatrix}\text{, and }B =\begin{bmatrix} 0 \\ 0 \\ 1 \\ \frac1d \end{bmatrix}.\end{split}$

Functional equation (12) has solution

$v(y_t) = -y_t'Py_t$

where

• $$P$$ solves the algebraic matrix Riccati equation $$P = R+ \beta A'PA- \beta A'PB(B'PB)^{-1}B'PA$$
• the optimal policy function is given by $$\tau_{t+1} = -F y_t$$ for $$F = (B'PB)^{-1}B'PA$$

Now we turn to subproblem 1.

Evidently the optimal choice of $$u_0$$ satisfies $$\frac{\partial v}{\partial u_0} =0$$

If we partition $$P$$ as

$\begin{split}P = \begin{bmatrix} P_{11}&P_{12} \\ P_{21}&P_{22} \end{bmatrix}\end{split}$

then we have

$0 = \frac{\partial}{\partial u_0} \left( z_0'P_{11}z_0+z_0'P_{12}u_0+u_0'P_{21}z_0 +u_0' P_{22} u_0 \right) = P_{12}'z_0+P_{21}z_0+2P_{22}u_0$

which implies

(13)$u_0 = -P_{22}^{-1}P_{21}z_0$

Thus, the Ramsey plan is

$\begin{split}\tau_{t+1} = -F \begin{bmatrix} z_t \\ u_t \end{bmatrix} \quad \text{and} \quad \begin{bmatrix} z_{t+1} \\ u_{t+1} \end{bmatrix} = (A-BF) \begin{bmatrix} z_t \\ u_t \end{bmatrix}\end{split}$

with initial state $$\begin{bmatrix} z_0 & -P_{22}^{-1}P_{21}z_0\end{bmatrix}'$$

### Recursive Representation¶

An outcome of the preceding results is that the Ramsey plan can be represented recursively as the choice of an initial marginal utility (or rate of growth of output) according to a function

(14)$u_0 = \upsilon(Q_0|\mu)$

that obeys (13) and the following updating equations for $$t\geq 0$$:

(15)$\tau_{t+1} = \tau(Q_t, u_t|\mu)$
(16)$Q_{t+1} = Q_t + u_t$
(17)$u_{t+1} = u(Q_t, u_t|\mu)$

We have conditioned the functions $$\upsilon$$, $$\tau$$, and $$u$$ by $$\mu$$ to emphasize how the dependence of $$F$$ on $$G_0$$ appears indirectly through the Lagrange multiplier $$\mu$$

### An Example Calculation¶

We’ll discuss how to compute $$\mu$$ below but first consider the following numerical example

We take the parameter set $$[A_0, A_1, d, \beta, Q_0] = [100, .05, .2, .95, 100]$$ and compute the Ramsey plan with the following piece of code

#=

@author : Spencer Lyon <spencer.lyon@nyu.edu>
Victoria Gregory <victoria.gregory@nyu.edu>

=#
using QuantEcon
using Roots
using Plots
using LaTeXStrings
pyplot()

struct HistDepRamsey{TF<:AbstractFloat}
# These are the parameters of the economy
A0::AbstractFloat
A1::AbstractFloat
d::AbstractFloat
Q0::AbstractFloat
τ0::AbstractFloat
μ0::AbstractFloat
β::AbstractFloat

# These are the LQ fields and stationary values
R::Matrix{TF}
A::Matrix{TF}
B::Vector{TF}
Q::TF
P::Matrix{TF}
F::Matrix{TF}
lq::LQ
end

struct RamseyPath{TF<:AbstractFloat}
y::Matrix{TF}
uhat::Vector{TF}
uhatdif::Vector{TF}
τhat::Vector{TF}
τhatdif::Vector{TF}
μ::Vector{TF}
G::Vector{TF}
GPay::Vector{TF}
end

function HistDepRamsey(A0::AbstractFloat,
A1::AbstractFloat,
d::AbstractFloat,
Q0::AbstractFloat,
τ0::AbstractFloat,
μ::AbstractFloat,
β::AbstractFloat)
# Create Matrices for solving Ramsey problem

R = [     0.0  -A0 / 2      0.0      0.0
-A0 / 2    A1 / 2   -μ / 2      0.0
0.0    -μ / 2      0.0      0.0
0.0       0.0      0.0    d / 2]

A = [    1.0      0.0    0.0               0.0
0.0      1.0    0.0               1.0
0.0      0.0    0.0               0.0
-A0 / d   A1 / d    0.0  A1 / d + 1.0 / β]

B = [0.0; 0.0; 1.0; 1.0 / d]

Q = 0.0

# Use LQ to solve the Ramsey Problem.
lq = LQ(Q, -R, A, B, bet=β)

P, F, _d = stationary_values(lq)

HistDepRamsey(A0, A1, d, Q0, τ0, μ0, β, R, A, B, Q, P, Array(F), lq)
end

function compute_G(hdr::HistDepRamsey, μ::AbstractFloat)
# simplify notation
Q0, τ0, A, B, Q = hdr.Q0, hdr.τ0, hdr.A, hdr.B, hdr.Q
β = hdr.β

R = hdr.R
R[2, 3] = R[3, 2] = -μ / 2
lq = LQ(Q, -R, A, B, bet=β)

P, F, _d = stationary_values(lq)

# Need y_0 to compute government tax revenue.
u0 = compute_u0(hdr, P)
y0 = vcat([1.0 Q0 τ0]', u0)

# Define A_F and S matricies
AF = A - B * F
S = [0.0 1.0 0.0 0]' * [0.0 0.0 1.0 0]

# Solves equation (25)
Ω = solve_discrete_lyapunov(sqrt(β) .* AF', β * AF' * S * AF)
T0 = y0' * Ω * y0

return T0[1], A, B, F, P
end

function compute_u0(hdr::HistDepRamsey, P::Matrix)
# simplify notation
Q0, τ0 = hdr.Q0, hdr.τ0

P21 = P[4, 1:3]
P22 = P[4, 4]
z0 = [1.0 Q0 τ0]'
u0 = reshape(-P22 ^ (-1) .* P21, 1, 3) * (z0)

return u0[1]
end

function init_path{TF<:AbstractFloat}(hdr::HistDepRamsey{TF}, μ0::TF, T::Integer=20)
# Construct starting values for the path of the Ramsey economy
G0, A, B, F, P = compute_G(hdr, μ0)

# Compute the optimal u0
u0 = compute_u0(hdr, P)

# Initialize vectors
y = Array{TF}(4, T)
uhat = Vector{TF}(T)
uhatdif = Vector{TF}(T)
τhat = Vector{TF}(T)
τhatdif = Vector{TF}(T)
μ = Vector{TF}(T)
G = Vector{TF}(T)
GPay = Vector{TF}(T)

# Initial conditions
G[1] = G0
μ[1] = μ0
uhatdif[1] = 0
uhat[1] = u0
τhatdif[1] = 0
y[:, 1] = vcat([1.0 hdr.Q0 hdr.τ0]', u0)

return RamseyPath(y, uhat, uhatdif, τhat, τhatdif, μ, G, GPay)
end

function compute_ramsey_path!(hdr::HistDepRamsey, rp::RamseyPath)
# simplify notation
y, uhat, uhatdif, τhat, = rp.y, rp.uhat, rp.uhatdif, rp.τhat
τhatdif, μ, G, GPay = rp.τhatdif, rp.μ, rp.G, rp.GPay
β = hdr.β

_, A, B, F, P = compute_G(hdr, μ[1])

for t=2:T
# iterate government policy
y[:, t] = (A - B * F) * y[:, t-1]

# update G
G[t] = (G[t - 1] - β * y[2, t] * y[3, t]) / β
GPay[t] = β .* y[2, t] * y[3, t]

#=
Compute the μ if the government were able to reset its plan
ff is the tax revenues the government would receive if they reset the
plan with Lagrange multiplier μ minus current G
=#
ff(μ) = compute_G(hdr, μ)[1] - G[t]

# find ff = 0
μ[t] = fzero(ff, μ[t - 1])
_, Atemp, Btemp, Ftemp, Ptemp = compute_G(hdr, μ[t])

# Compute alternative decisions
P21temp = Ptemp[4, 1:3]
P22temp = P[4, 4]
uhat[t] = dot(-P22temp ^ (-1) .* P21temp, y[1:3, t])

yhat = (Atemp-Btemp * Ftemp) * [y[1:3, t-1]; uhat[t-1]]
τhat[t] = yhat[4]
τhatdif[t] = τhat[t] - y[4, t]
uhatdif[t] = uhat[t] - y[4, t]
end

return rp
end

# Primitives
T = 20
A0 = 100.0
A1 = 0.05
d = 0.20
β = 0.95

# Initial conditions
μ0 = 0.0025
Q0 = 1000.0
τ0 = 0.0

# Solve Ramsey problem and compute path
hdr = HistDepRamsey(A0, A1, d, Q0, τ0, μ0, β)
rp = init_path(hdr, μ0, T)
compute_ramsey_path!(hdr, rp);                 # updates rp in place


The above code computes a number of sequences besides the Ramsey plan, some of which have already been discussed, while others will be described below

The next figure uses the program to compute and show the Ramsey plan for $$\tau$$ and the Ramsey outcome for $$(Q_t,u_t)$$

function plot1(rp::RamseyPath)
T = length(rp.μ)
y = rp.y

ylabels = [L"$Q$" L"$\tau$" L"$u$"]
y_vals = [y[2, :] y[3, :] y[4, :]]
p = plot(0:T, y_vals, color=:blue,
label=["Output" "Tax rate" "First difference in output"],
size=(600, 600), lw=2, alpha=0.7, ylabel=ylabels, layout=(3, 1),
xlims=(0, 15), xlabel=["" "" "Time"], legend=:topright,
xticks=0:5:15)
return p
end

plot1(rp)


From top to bottom, the panels show $$Q_t$$, $$\tau_t$$ and $$u_t := Q_{t+1} - Q_t$$ over $$t=0, \ldots, 15$$

The optimal decision rule is [4]

(18)$\tau_{t+1} = -248.0624 - 0.1242 Q_t - 0.3347 u_t$

Notice how the Ramsey plan calls for a high tax at $$t=1$$ followed by a perpetual stream of lower taxes

Taxing heavily at first, less later expresses time-inconsistency of the optimal plan for $$\{\tau_{t+1}\}_{t=0}^\infty$$

We’ll characterize this formally after first discussing how to compute $$\mu$$.

### Computing $$\mu$$¶

Define the selector vectors $$e_\tau = \begin{bmatrix} 0 & 0 & 1 & 0 \end{bmatrix}'$$ and $$e_Q = \begin{bmatrix} 0 & 1 & 0 & 0 \end{bmatrix}'$$ and express $$\tau_t = e_\tau' y_t$$ and $$Q_t = e_Q' y_t$$

Evidently $$Q_t \tau_t = y_t' e_Q e_\tau' y_t = y_t' S y_t$$ where $$S := e_Q e_\tau'$$

We want to compute

$T_0 = \sum_{t=1}^\infty \beta^t \tau_t Q_t = \tau_1 Q_1 + \beta T_1$

where $$T_1 = \sum_{t=2}^\infty \beta^{t-1} Q_t \tau_t$$

The present values $$T_0$$ and $$T_1$$ are connected by

$T_0 = \beta y_0' A_F' S A_F y_0 + \beta T_1$

Guess a solution that takes the form $$T_t = y_t' \Omega y_t$$, then find an $$\Omega$$ that satisfies

(19)$\Omega = \beta A_F' S A_F + \beta A_F' \Omega A_F$

Equation (19) is a discrete Lyapunov equation that can be solved for $$\Omega$$ using QuantEcon’s solve_discrete_lyapunov function

The matrix $$F$$ and therefore the matrix $$A_F = A-BF$$ depend on $$\mu$$

To find a $$\mu$$ that guarantees that $$T_0 = G_0$$ we proceed as follows:

1. Guess an initial $$\mu$$, compute a tentative Ramsey plan and the implied $$T_0 = y_0' \Omega(\mu) y_0$$
2. If $$T_0 > G_0$$, lower $$\mu$$; otherwise, raise $$\mu$$
3. Continue iterating on step 3 until $$T_0 = G_0$$

## Time Inconsistency¶

Recall that the Ramsey planner chooses $$\{u_t\}_{t=0}^\infty, \{\tau_t\}_{t=1}^\infty$$ to maximize

$\sum_{t=0}^\infty \beta^t \left\{ A_0 Q_t - \frac{A_1}{2}Q_t^2 - \frac{d}{2} u_t^2 \right\}$

subject to (4), (5), and (7)

We express the outcome that a Ramsey plan is time-inconsistent the following way

Proposition. A continuation of a Ramsey plan is not a Ramsey plan

Let

(20)$w(Q_0,u_0|\mu_0) = \sum_{t=0}^\infty \beta^t \left\{ A_0 Q_t - \frac{A_1}{2}Q_t^2 - \frac{d}{2} u_t^2 \right\}$

where

• $$\{Q_t,u_t\}_{t=0}^\infty$$ are evaluated under the Ramsey plan whose recursive representation is given by (15), (16), (17)
• $$\mu_0$$ is the value of the Lagrange multiplier that assures budget balance, computed as described above

Evidently, these continuation values satisfy the recursion

(21)$w(Q_t,u_t|\mu_0) = A_0 Q_{t} - \frac{A_1}{2} Q_{t}^2 - \frac{d}{2} u_{t}^2 + \beta w (Q_{t+1},u_{t+1}|\mu_0)$

for all $$t \geq 0$$, where $$Q_{t+1} = Q_t + u_t$$

Under the timing protocol affiliated with the Ramsey plan, the planner is committed to the outcome of iterations on (15), (16), (17)

In particular, when time $$t$$ comes, the Ramsey planner is committed to the value of $$u_t$$ implied by the Ramsey plan and receives continuation value $$w(Q_t,u_t,\mu_0)$$

That the Ramsey plan is time-inconsistent can be seen by subjecting it to the following ‘revolutionary’ test

First, define continuation revenues $$G_t$$ that the government raises along the original Ramsey outcome by

(22)$G_t = \beta^{-t}(G_0-\sum_{s=1}^t\beta^s\tau_sQ_s)$

where $$\{\tau_t, Q_t\}_{t=0}^\infty$$ is the original Ramsey outcome [5]

Then at time $$t \geq 1$$,

1. take $$(Q_t, G_t)$$ inherited from the original Ramsey plan as initial conditions
2. invite a brand new Ramsey planner to compute a new Ramsey plan, solving for a new $$u_t$$, to be called $${\check u_t}$$, and for a new $$\mu$$, to be called $${\check \mu_t}$$

The revised Lagrange multiplier $$\check{\mu_t}$$ is chosen so that, under the new Ramsey plan, the government is able to raise enough continuation revenues $$G_t$$ given by (22)

Would this new Ramsey plan be a continuation of the original plan?

The answer is no because along a Ramsey plan, for $$t \geq 1$$, in general it is true that

(23)$w\bigl(Q_t, \upsilon(Q_t|\check{\mu})|\check{\mu}\bigr) > w(Q_t, u_t|\mu_0)$

Inequality (23) expresses a continuation Ramsey planner’s incentive to deviate from a time $$0$$ Ramsey plan by

1. resetting $$u_t$$ according to (14)
2. adjusting the Lagrange multiplier on the continuation appropriately to account for tax revenues already collected [6]

Inequality (23) expresses the time-inconsistency of a Ramsey plan

### A Simulation¶

To bring out the time inconsistency of the Ramsey plan, we compare

• the time $$t$$ values of $$\tau_{t+1}$$ under the original Ramsey plan with
• the value $$\check \tau_{t+1}$$ associated with a new Ramsey plan begun at time $$t$$ with initial conditions $$(Q_t, G_t)$$ generated by following the original Ramsey plan

Here again $$G_t := \beta^{-t}(G_0-\sum_{s=1}^t\beta^s\tau_sQ_s)$$

The difference $$\Delta \tau_t := \check{\tau_t} - \tau_t$$ is shown in the top panel of the following figure

function plot2(rp::RamseyPath)
y, uhatdif, τhatdif, μ = rp.y, rp.uhatdif, rp.τhatdif, rp.μ
G, GPay = rp.G, rp.GPay
T = length(rp.μ)

y_vals = [τhatdif uhatdif μ G]
ylabels = [L"$\Delta\tau$" L"$\Delta u$" L"$\mu$" L"$G$"]
labels = hcat("Time inconsistency differential for tax rate",
L"Time inconsistency differential for $u$",
"Lagrange multiplier", "Government revenue")
p = plot(0:T , y_vals, ylabel=ylabels, label=labels, size=(800, 800),
lw=2, layout=(4, 1), xlims=(-0.5, 15), alpha=0.7,
legend=:topright, color=:blue, xlabel=["" "" "" "Time"])
return p
end

plot2(rp)


In the second panel we compare the time $$t$$ outcome for $$u_t$$ under the original Ramsey plan with the time $$t$$ value of this new Ramsey problem starting from $$(Q_t, G_t)$$

To compute $$u_t$$ under the new Ramsey plan, we use the following version of formula (13):

$\check{u_t} = - P_{22}^{-1} (\check\mu_{t}) P_{21}(\check\mu_t) z_t$

Here $$z_t$$ is evaluated along the Ramsey outcome path, where we have included $$\check{\mu_t}$$ to emphasize the dependence of $$P$$ on the Lagrange multiplier $$\mu_0$$ [7]

To compute $$u_t$$ along the Ramsey path, we just iterate the recursion starting (17) from the initial $$Q_0$$ with $$u_0$$ being given by formula (13)

Thus the second panel indicates how far the reinitialized value $$\check{u_t}$$ value departs from the time $$t$$ outcome along the Ramsey plan

Note that the restarted plan raises the time $$t+1$$ tax and consequently lowers the time $$t$$ value of $$u_t$$

Associated with the new Ramsey plan at $$t$$ is a value of the Lagrange multiplier on the continuation government budget constraint

This is the third panel of the figure

The fourth panel plots the required continuation revenues $$G_t$$ implied by the original Ramsey plan

These figures help us understand the time inconsistency of the Ramsey plan

#### Further Intuition¶

One feature to note is the large difference between $$\check \tau_{t+1}$$ and $$\tau_{t+1}$$ in the top panel of the figure

If the government is able to reset to a new Ramsey plan at time $$t$$, it chooses a significantly higher tax rate than if it were required to maintain the original Ramsey plan

The intuition here is that the government is required to finance a given present value of expenditures with distorting taxes $$\tau$$

The quadratic adjustment costs prevent firms from reacting strongly to variations in the tax rate for next period, which tilts a time $$t$$ Ramsey planner toward using time $$t+1$$ taxes

As was noted before, this is evident in the first figure, where the government taxes the next period heavily and then falls back to a constant tax from then on

This can also been seen in the third panel of the second figure, where the government pays off a significant portion of the debt using the first period tax rate

The similarities between the graphs in the last two panels of the second figure reveals that there is a one-to-one mapping between $$G$$ and $$\mu$$

The Ramsey plan can then only be time consistent if $$G_t$$ remains constant over time, which will not be true in general

## Credible Policy¶

We express the theme of this section in the following: In general, a continuation of a Ramsey plan is not a Ramsey plan

This is sometimes summarized by saying that a Ramsey plan is not credible

On the other hand, a continuation of a credible plan is a credible plan

The literature on a credible public policy ([CK90] and [Sto89]) arranges strategies and incentives so that public policies can be implemented by a sequence of government decision makers instead of a single Ramsey planner who chooses an entire sequence of history-dependent actions once and for all at time $$t=0$$

Here we confine ourselves to sketching how recursive methods can be used to characterize credible policies in our model

A key reference on these topics is [Cha98]

A credibility problem arises because we assume that the timing of decisions differs from those for a Ramsey problem

A sequential timing protocol is a protocol such that

1. At each $$t \geq 0$$, given $$Q_t$$ and expectations about a continuation tax policy $$\{\tau_{s+1}\}_{s=t}^\infty$$ and a continuation price sequence $$\{p_{s+1}\}_{s=t}^\infty$$, the representative firm chooses $$u_t$$
2. At each $$t$$, given $$(Q_t, u_t)$$, a government chooses $$\tau_{t+1}$$

Item (2) captures that taxes are now set sequentially, the time $$t+1$$ tax being set after the government has observed $$u_t$$

Of course, the representative firm sets $$u_t$$ in light of its expectations of how the government will ultimately choose to set future taxes

A credible tax plan $$\{\tau_{s+1}\}_{s=t}^\infty$$

• is anticipated by the representative firm, and
• is one that a time $$t$$ government chooses to confirm

We use the following recursion, closely related to but different from (21), to define the continuation value function for the government:

(24)$J_t = A_0 Q_{t} - \frac{A_1}{2} Q_{t}^2 - \frac{d}{2} u_{t}^2 + \beta J_{t+1} (\tau_{t+1},G_{t+1})$

This differs from (21) because

• continuation values are now allowed to depend explicitly on values of the choice $$\tau_{t+1}$$, and
• continuation government revenue to be raised $$G_{t+1}$$ need not be ones called for by the prevailing government policy

Thus, deviations from that policy are allowed, an alteration that recognizes that $$\tau_t$$ is chosen sequentially

Express the government budget constraint as requiring that $$G_0$$ solves the difference equation

(25)$G_t = \beta \tau_{t+1} Q_{t+1} + \beta G_{t+1}, \quad t \geq 0$

subject to the terminal condition $$\lim_{t \rightarrow + \infty} \beta^t G_t= 0$$

Because the government is choosing sequentially, it is convenient to

• take $$G_t$$ as a state variable at $$t$$ and
• to regard the time $$t$$ government as choosing $$(\tau_{t+1}, G_{t+1})$$ subject to constraint (25)

To express the notion of a credible government plan concisely, we expand the strategy space by also adding $$J_t$$ itself as a state variable and allowing policies to take the following recursive forms [8]

Regard $$J_0$$ as an a discounted present value promised to the Ramsey planner and take it as an initial condition.

Then after choosing $$u_0$$ according to

(26)$u_0 = \upsilon(Q_0, G_0, J_0),$

choose subsequent taxes, outputs, and continuation values according to recursions that can be represented as

(27)$\hat \tau_{t+1} = \tau(Q_t, u_t, G_t, J_t )$
(28)$u_{t+1} = \xi (Q_t, u_t, G_t, J_t,{\tau_{t+1}} )$
(29)$G_{t+1} = \beta^{-1} G_t - \tau_{t+1} Q_{t+1}$
(30)$J_{t+1}(\tau_{t+1}, G_{t+1}) = \nu(Q_t, u_t, G_{t+1}, J_t, \tau_{t+1} )$

Here

• $$\hat \tau_{t+1}$$ is the time $$t+1$$ government action called for by the plan, while
• $$\tau_{t+1}$$ is possibly some one-time deviation that the time $$t+1$$ government contemplates and
• $$G_{t+1}$$ is the associated continuation tax collections

The plan is said to be credible if, for each $$t$$ and each state $$(Q_t, u_t, G_t, J_t)$$, the plan satisfies the incentive constraint

(31)\begin{split}\begin{aligned} J_t & = A_0 Q_{t} - \frac{A_1}{2} Q_{t}^2 - \frac{d}{2} u_{t}^2 + \beta J_{t+1} (\hat \tau_{t+1}, \hat G_{t+1}) \\ & \geq A_0 Q_{t} - \frac{A_1}{2} Q_{t}^2 - \frac{d}{2} u_{t}^2 + \beta J_{t+1} ( \tau_{t+1}, G_{t+1}) \end{aligned}\end{split}

for all tax rates $$\tau_{t+1} \in {\mathbb R}$$ available to the government

Here $$\hat G_{t+1} = \frac{G_t - \hat \tau_{t+1} Q_{t+1}}{\beta}$$

• Inequality expresses that continuation values adjust to deviations in ways that discourage the government from deviating from the prescribed $$\hat \tau_{t+1}$$
• Inequality (31) indicates that two continuation values $$J_{t+1}$$ contribute to sustaining time $$t$$ promised value $$J_t$$
• $$J_{t+1} (\hat \tau_{t+1}, \hat G_{t+1})$$ is the continuation value when the government chooses to confirm the private sector’s expectation, formed according to the decision rule (27) [9]
• $$J_{t+1}(\tau_{t+1}, G_{t+1})$$ tells the continuation consequences should the government disappoint the private sector’s expectations

The internal structure of a credible plan deters deviations from it

That (31) maps two continuation values $$J_{t+1}(\tau_{t+1},G_{t+1})$$ and $$J_{t+1}(\hat \tau_{t+1},\hat G_{t+1})$$ into one promised value $$J_t$$ reflects how a credible plan arranges a system of private sector expectations that induces the government to choose to confirm them

Chang [Cha98] builds on how inequality (31) maps two continuation values into one

Remark Let $${\mathcal J}$$ be the set of values associated with credible plans

Every value $$J \in {\mathcal J}$$ can be attained by a credible plan that has a recursive representation of form form (27), (28), (29)

The set of values can be computed as the largest fixed point of an operator that maps sets of candidate values into sets of values

Given a value within this set, it is possible to construct a government strategy of the recursive form (27), (28), (29) that attains that value

In many cases, there is a set of values and associated credible plans

In those cases where the Ramsey outcome is credible, a multiplicity of credible plans is a key part of the story because, as we have seen earlier, a continuation of a Ramsey plan is not a Ramsey plan

For it to be credible, a Ramsey outcome must be supported by a worse outcome associated with another plan, the prospect of reversion to which sustains the Ramsey outcome

## Concluding remarks¶

The term ‘optimal policy’, which pervades an important applied monetary economics literature, means different things under different timing protocols

Under the ‘static’ Ramsey timing protocol (i.e., choose a sequence once-and-for-all), we obtain a unique plan

Here the phrase ‘optimal policy’ seems to fit well, since the Ramsey planner optimally reaps early benefits from influencing the private sector’s beliefs about the government’s later actions

When we adopt the sequential timing protocol associated with credible public policies, ‘optimal policy’ is a more ambiguous description

There is a multiplicity of credible plans

True, the theory explains how it is optimal for the government to confirm the private sector’s expectations about its actions along a credible plan

But some credible plans have very bad outcomes

These bad outcomes are central to the theory because it is the presence of bad credible plans that makes possible better ones by sustaining the low continuation values that appear in the second line of incentive constraint (31)

Recently, many have taken for granted that ‘optimal policy’ means ‘follow the Ramsey plan’ [10]

In pursuit of more attractive ways to describe a Ramsey plan when policy making is in practice done sequentially, some writers have repackaged a Ramsey plan in the following way

• Take a Ramsey outcome - a sequence of endogenous variables under a Ramsey plan - and reinterpret it (or perhaps only a subset of its variables) as a target path of relationships among outcome variables to be assigned to a sequence of policy makers [11]
• If appropriate (infinite dimensional) invertibility conditions are satisfied, it can happen that following the Ramsey plan is the only way to hit the target path [12]
• The spirit of this work is to say, “in a democracy we are obliged to live with the sequential timing protocol, so let’s constrain policy makers’ objectives in ways that will force them to follow a Ramsey plan in spite of their benevolence” [13]
• By this slight of hand, we acquire a theory of an optimal outcome target path

This ‘invertibility’ argument leaves open two important loose ends:

1. implementation, and
2. time consistency

As for (1), repackaging a Ramsey plan (or the tail of a Ramsey plan) as a target outcome sequence does not confront the delicate issue of how that target path is to be implemented [14]

As for (2), it is an interesting question whether the ‘invertibility’ logic can repackage and conceal a Ramsey plan well enough to make policy makers forget or ignore the benevolent intentions that give rise to the time inconsistency of a Ramsey plan in the first place

To attain such an optimal output path, policy makers must forget their benevolent intentions because there will inevitably occur temptations to deviate from that target path, and the implied relationship among variables like inflation, output, and interest rates along it

Remark The continuation of such an optimal target path is not an optimal target path

Footnotes

 [1] We could also call a competitive equilibrium a rational expectations equilibrium.
 [2] It is important not to set $$q_t = Q_t$$ prematurely. To make the firm a price taker, this equality should be imposed after and not before solving the firm’s optimization problem.
 [3] We could instead, perhaps with more accuracy, define a promised marginal value as $$\beta (A_0 - A_1 Q_{t+1} ) - \beta \tau_{t+1} + u_{t+1}/\beta$$, since this is the object to which the firm’s first-order condition instructs it to equate to the marginal cost $$d u_t$$ of $$u_t = q_{t+1} - q_t$$. This choice would align better with how Chang [Cha98] chose to express his competitive equilibrium recursively. But given $$(u_t, Q_t)$$, the representative firm knows $$(Q_{t+1},\tau_{t+1})$$, so it is adequate to take $$u_{t+1}$$ as the intermediate variable that summarizes how $$\vec \tau_{t+1}$$ affects the firm’s choice of $$u_t$$.
 [4] As promised, $$\tau_t$$ does not appear in the Ramsey planner’s decision rule for $$\tau_{t+1}$$.
 [5] The continuation revenues $$G_t$$ are the time $$t$$ present value of revenues that must be raised to satisfy the original time $$0$$ government intertemporal budget constraint, taking into account the revenues already raised from $$s=1, \ldots, t$$ under the original Ramsey plan.
 [6] For example, let the Ramsey plan yield time $$1$$ revenues $$Q_1 \tau_1$$. Then at time $$1$$, a continuation Ramsey planner would want to raise continuation revenues, expressed in units of time $$1$$ goods, of $$\tilde G_1 := \frac{G - \beta Q_1 \tau_1}{\beta}$$. To finance the remainder revenues, the continuation Ramsey planner would find a continuation Lagrange multiplier $$\mu$$ by applying the three-step procedure from the previous section to revenue requirements $$\tilde G_1$$.
 [7] It can be verified that this formula puts non-zero weight only on the components $$1$$ and $$Q_t$$ of $$z_t$$.
 [8] This choice is the key to what [LS18] call ‘dynamic programming squared’.
 [9] Note the double role played by (27): as decision rule for the government and as the private sector’s rule for forecasting government actions.
 [10] It is possible to read [Woo03] and [GW10] as making some carefully qualified statements of this type. Some of the qualifications can be interpreted as advice ‘eventually’ to follow a tail of a Ramsey plan.
 [11] In our model, the Ramsey outcome would be a path $$(\vec p, \vec Q)$$.
 [12] See [GW10].
 [13] Sometimes the analysis is framed in terms of following the Ramsey plan only from some future date $$T$$ onwards.
 [14] See [Bas05] and [ACK10].
• Share page